
ABSTRACT

Beginning SAS® programmers soon encounter the problem of showing in a summary report all possible categorical values even when not all values are in the data.  This tutorial paper presents the SAS code for two different approaches to a solution.

INTRODUCTION

This paper is a tutorial for beginning SAS programmers.  It describes a common problem and how it can arise.  It then addresses the problem by presenting the code for two very different approaches.  The first approach consists of mostly data step programming.   The second approach relies on frequently used base SAS procedures and what at first appears as a trick.  Both approaches, however, are frequently used. 

THE DATA AND THE PROBLEM

To simplify the discussion and the presentation, a datalines statement in a data step creates a small example dataset, DS.

DATASET: DS        ID        DESC
                   10        G

                   20        G

                   30         

                   40        B

                   50        G

The following code creates the associated summary dataset, SUMDS.

CODE    proc freq data=DS;

               tables DESC / out= SUMDS  (keep=DESC COUNT); 

        label COUNT=’COUNT’; run;

DATASET: SUMDS     DESC        COUNT

                               1

                   B           1

                   G           3

I first encountered this specific problem in a maintenance issue.  The user claimed ‘the previous summary report had all the information but now some values are missed on the current summary report’.  Specifically, he needs to see values even when the associated count is zero.  I soon learn that the list of all possible values for the DESC variable is (B, G, <blank>, U).  The problem with dataset SUMDS, is that there is no row with DESC = ‘U’ and COUNT = 0.  But the user expects and requires such a row. 

SOLUTION  IDEAS

The first solution uses, as much as possible, data step coding and datasets whereas the second solution uses the power of various procedures.  During analysis, it is noticed that to solve the problem, the system must remember from one session to another the list of all possible values of the DESC variable.   To achieve this, the first solution uses a datalines statement to build a dataset, ALLDESC that has all possible values of the DESC variable.  However, the second solution uses a user-defined format and an associated control dataset to build ALLDESC. 

DATASET: ALLDESC   DESC
                   <blank>

                   B 

                   G 

                   U 

SOLUTION ONE

Solution one is mostly data step programming, uses grouping logic,  and produces a summary dataset, SUM1.   We can retain and increment using only ‘COUNT + 1’.  

CODE    data ALLDESC;

                 input DESC $ 1;

                        datalines;

                   G

                   B

                  U

                  ;

                  run;

       proc sort data= ALLDESC;

          by DESC; run;

       proc sort data= DS;

          by DESC; run;

       data DescID;

          merge ALLDESC  DS;

          by DESC; run;

       data SUM1  (keep= DESC  COUNT);
          set DescID;

          by DESC;

          if first.DESC then COUNT = 0;

          if ID ^= ' '  then COUNT + 1;

          if last.DESC  then output; run;

DATASET: DescID    DESC        ID
                               30

                   B           40

                   G           10

                   G           20

                   G           50

                   U           

DATASET: SUM1      DESC        COUNT
                               1

                   B           1

                   G           3

                   U           0

SOLUTION TWO

The second approach relies on what initially appears as a trick.  However, this approach is just another application of the ‘template’ concept.  Briefly, we create a blank or dummy template dataset that has all the DESC values needed and has the rest of the cells ‘zero-ed’ out.  Then, we appropriately overwrite the individual zero cells with COUNT values.  

First, use a user-defined format and control a dataset to build the dataset, ALLDESC, of all possible DESC values. 

CODE    proc format;   value $gbu   

          G   = "Good Effort"

          B   = "Brilliant Work"

          U   = "Unsatisfactory"

          ' ' = "Missing" ; quit; run;

        proc format CNTLOUT = ALLDESC;

          select $gbu; quit; run;

        proc sort data = ALLDESC (rename=(start=DESC))

          out  = ALLDESC (keep= DESC); label DESC=’DESC’; 

          by DESC; run;

Use the dataset, ALLDESC, to create the template dataset, DESCTEMPLATE.  The corresponding code follows.  

CODE    data DESCTEMPLATE;

          set ALLDESC; 
          COUNT=0; run;

DATASET: DESCTEMPLATE  DESC        COUNT
                                   0

                       B           0

                       G           0

                       U           0

Next, run the above proc freq that creates dataset, SUMDS.  Finally, merge or update these datasets to appropriately overwrite the zero-cells in the template dataset.  The code that produces a summary dataset SUM2 similar to SUM1 (above) follows. 

CODE    proc freq data=DS;

                 tables DESC / out= SUMDS  (keep=DESC COUNT); 

          label COUNT=’COUNT’; run;

                  data SUM2  (drop=DESC);
          update DESCTEMPLATE SUMDS; 

          by DESC; run;

ENHANCEMENTS TO SOLUTION ONE

We can enhance the reporting of the summary datasets by displaying values:  ‘Missing’, ‘Brilliant Work, ‘Good Effort’, and ‘Unsatisfactory’ instead of the actual variable values.   For solution one, during the construction of ALLDESC another variable, LABEL, can be included with values equal to:  ‘Missing’, ‘Brilliant Work, ‘Good Effort’, and ‘Unsatisfactory’.  This variable can then ride along and be used as a sort and display variable.

Another possibility  is to apply the format $gbu to the DESC variable in dataset SUM1 producing SUM1e. Then to order the dataset rows by label values, you could use proc report.

CODE    proc report data=SUM1e nowd missing out=S1f;

           column DESC COUNT;

           define DESC  / 'DESC'  order;

           define COUNT / 'COUNT' display center; run;

ENHANCEMENTS TO SOLUTION TWO

For the second solution, in the steps that create ALLDESC consider keeping the automatically created variable LABEL and then later using it as a sort and display variable.  To do this, change a previous proc sort call to read: 

CODE    proc sort data = ALLDESC (rename=(start=DESC label=’DESCRIP’))

           out  = ALLDESC (keep=DESC DESCRIP);

           label  DESC=’DESC’  DESCRIP=’DESCRIP’;

           by DESC; run;

CONCLUSION

Beginners doing either SAS development or maintenance would benefit by knowing both the ‘mostly data step’ programmatic solution and the commonly used ‘template/proc’ approach. 

APPENDIX:  SAS CODE FOR THE SOLUTIONS AND ENHANCEMENTS

/*Each solution and enhancement can stand on its own without dependence on output produced by earlier code.*/

/*##### THE DATA AND THE PROBLEM  #########################################*/

data ds;

  input ID $1-2 DESC $ 4 ;

  datalines;

10 G

20 G

30 

40 B

50 G

;

run;

proc freq data=DS;

  tables desc / out= SUMDS  (keep=DESC COUNT); 

  label COUNT=’COUNT’; run;

/*#####  CODE FOR SOLUTION ONE  #############################################*/

data ds;

  input ID $1-2 DESC $ 4 ;

  datalines;

10 G

20 G

30 

40 B

50 G

;

run;

data ALLDESC;

  input DESC $ 1;

  datalines;

G

B

U

;

run;

proc sort data=ALLDESC;

  by DESC; run;

proc sort data=DS;

  by DESC; run;

data DescID;

  merge ALLDESC  DS;

  by DESC; run;

data SUM1 (keep=DESC COUNT);

  set DescID;

  by DESC;

  if first.DESC then COUNT = 0;

  if ID ^= ' '  then COUNT + 1;

  if last.DESC  then output; run;

/*#####  CODE FOR SOLUTION TWO  ##############################################*/

/*The code for the second approach uses a user-defined format to 'store'

  all possible categorical variable values. The process of creating the 

  control dataset from the format puts the DESC values into a column 

  called START that we rename DESC.  The mixed case labels get placed 

  in a column called LABEL, that we do not use in this solution.  

  Finally, we sort by DESC in anticipation of a possible merge. */

data ds;

  input ID $1-2 DESC $ 4 ;

  datalines;

10 G

20 G

30 

40 B

50 G

;

run;

proc format;   value $gbu   

  G   = "Good Effort"

  B   = "Brilliant Work"

  U   = "Unsatisfactory"

  ' ' = "Missing" ; quit; run;

/*Using the format, create a control dataset 

  containing all possible category and label values*/

proc format CNTLOUT = ALLDESC;

  select $gbu; quit; run;

proc sort data = ALLDESC (rename=(start=DESC))

  out = ALLDESC (keep=DESC );  label DESC=’DESC’;

  by DESC; run;

data DESCTEMPLATE;

  set ALLDESC;

  COUNT=0; run;

proc freq data=DS;

  tables desc / out= SUMDS  (keep=DESC COUNT); 

  label COUNT=’COUNT’; run;

data SUM2;

  update DESCTEMPLATE SUMDS;

  by DESC; run;

/*#####  CODE FOR AN ENHANCEMENT TO SOLUTION ONE ##############################################*/

data ds;

  input ID $1-2 DESC $ 4 ;

  datalines;

10 G

20 G

30 

40 B

50 G

;

run;

data ALLDESC;

  input DESC $ 1;

  datalines;

G

B

U

;

run;

proc sort data=ALLDESC;

  by DESC; run;

proc sort data=DS;

  by DESC; run;

data DescID;

  merge ALLDESC  DS;

  by DESC; run;

data SUM1 (keep=DESC COUNT);

  set DescID;

  by DESC;

  if first.DESC then COUNT = 0;

  if ID ^= ' '  then COUNT + 1;

  if last.DESC  then output; run;

proc format;   value $gbu   

  G   = "Good Effort"

  B   = "Brilliant Work"

  U   = "Unsatisfactory"

  ' ' = "Missing" ; quit; run;

data SUM1e;

  set SUM1;

  format DESC $gbu. ; run;

proc report data=SUM1e nowd missing 

            out=SUM1f (keep=DESC  COUNT);

  column DESC COUNT;

  define DESC  / 'DESC'  order;

  define COUNT / 'COUNT' display center; run;

/*#####  CODE FOR AN ENHANCEMENT TO SOLUTION TWO ##############################################*/

data ds;

  input ID $1-2 DESC $ 4 ;

  datalines;

10 G

20 G

30 

40 B

50 G

;

run;

proc format;   value $gbu   

  G   = "Good Effort"

  B   = "Brilliant Work"

  U   = "Unsatisfactory"

  ' ' = "Missing" ; quit; run;

/*Using the format, create a control dataset 

  containing all possible category and label values*/

proc format CNTLOUT = ALLDESC;

  select $gbu; quit; run;

proc sort data = ALLDESC (rename=(start=DESC label=DESCRIP))

  out = ALLDESC (keep=DESC DESCRIP );

  label DESC=’DESC’  DESCRIP=’DESCRIP’;

  by DESC; run;

data DESCTEMPLATE ;

  set ALLDESC;

   COUNT=0; run;

proc freq data=DS ;

  tables DESC / out= SUMDS  (keep=DESC COUNT); 

  label COUNT=’COUNT’; run;

data SUM2 (drop=DESC);

  update DESCTEMPLATE SUMDS;

  label COUNT='COUNT';

  by DESC; run;

proc sort data = SUM2;

  by DESCRIP; run;

/*#####  END OF FILE  #################################################*/
